

Measuring the Parallax of Barnard's Star

Robert J. Vanderbei

Operations Research and Financial Engineering, Princeton University

rvdb@princeton.edu

ABSTRACT

Subject headings: Parallax, Barnard's Star, Parsec, Astronomical Unit

Barnard's Star is one of the closest stars to us. It is also the star that has the fastest apparent motion across the sky moving about 11 arcseconds per year. With a right ascension of 17h 53m 26s, it reaches opposition on the night of the summer solstice: June 21st.

Over the last two-plus years, I have taken six pictures of Barnard's star: on June 21 2012, June 6 2013, September 5 2013, April 10 2014, July 5, 2014, and October 27 2014. The September image is shown in Figure 1.

On each of the six nights, roughly 150 images were taken. Each image was an unguided 8-second exposure. For each night, roughly the best 80% of the images were selected and stacked to make a master image for the night. To maintain astrometric precision, the stacked images were saved in IEEE-float format fits files. Then, using MaximDL, the six images were aligned to each other using two nearby spectrally similar background stars for the alignment (see Figure 2). These aligned images were also saved in IEEE-float format. Again using MaximDL, the centroid of Barnard's Star was measured to a subpixel level. The centroids so measured turned out to be

Date	x-coord	y-coord
2012-06-21	681.527	639.539
2013-06-06	680.374	656.999
2013-09-05	678.639	661.234
2014-04-10	679.479	672.151
2014-07-05	677.950	676.775
2014-10-27	676.727	681.713

The six stacked and aligned images were then stacked with each other to show the relative motion of Barnard's Star. A closely cropped version of the final stack is shown in Figure 3. Also shown



Fig. 1.— A stack of the best 120 out of 150 8-second exposures taken on September 5, 2013, starting at 21:25 EST. Barnard's Star is the brightest star in the image. The telescope is a 10" RC from Ritchey-Chretien Optical Systems. The camera used is a Starlight Express SXV-H9.




Fig. 2.— The two stars chosen for the 2-star alignment in MaximDL. These two stars were chosen for their relative brightness, close proximity to Barnard’s star, and most importantly spectral similarity to Barnard’s star. Choosing spectrally similar stars obviated the need to correct for chromatic atmospheric distortion at the different altitudes encountered on the various nights.




Fig. 3.— After aligning the six images based on the background stars, the images were stacked and then closely cropped around Barnard’s Star. The cropped image was upsampled by a factor of 10. Also shown here is the best fit regression line showing the effect of parallax on the proper motion.

are the best-fit proper motion paths with and without accounting for parallax. These paths were computed as follows.

The parallax path is an ellipse whose semimajor axis is parallel to the ecliptic. Given that Barnard’s star’s position coincides with the winter solstice, it follows that the semimajor axis aligns with an east-west motion, which is essentially the horizontal direction in the images. Adding a linear motion to this elliptical motion, we get

$$x(t) = x_0 + v_x t + a \sin(2\pi t) \quad \text{and} \quad y(t) = y_0 + v_y t + b \cos(2\pi t)$$

where x_0 and y_0 represent the pixel values for $t = 0$, v_x and v_y represent the east-west and north-south rates of proper motion (in pixels per year), a is the semi-major axis of the parallax wobble (in pixels), b is the semi-minor axis of the parallax ellipse, and t is time measured in years with $t = 0$ corresponding to a conjunction (or opposition).

I chose $t = 0$ to correspond to the conjunction on December 21, 2012. Using the Sun’s right ascension at the times of acquisition of the six images, it is easy to determine t for each image. For example, at the first observation on June 21, 2012 at 3:21EDT, the Sun’s RA was 06h 01m 26s. Converting to decimal hours, this RA is $6 + (1 + 26/60)/60 = 6.024$. The sun’s RA at conjunction is the same as Barnard’s star’s RA, which in decimal format is $17 + (53 + 26/60)/60 = 17.890$. The time of the first observation, relative to the conjunction is obtain by subtracting the two RA’s: $6.024 - 17.890 = -11.866$. Of course, this number is in units of RA-hours. To convert to years, we simply need to divide by the number of RA-hours in a year to get $-11.866/24 = -0.4944$. Here’s the complete set a time calculations:

Date	Sun’s RA (hh:mm:ss)	Sun’s RA (decimal)	time (RA hours)	time (yrs)
2012-06-21	06:01:26	6.02	$6.02 - 17.89 = -11.87$	-0.494
2012-12-21	17:53:26	17.89	$17.89 - 17.89 = 0.0$	0.0
2013-06-06	04:58:02	4.97	$24 + 4.97 - 17.89 = 11.08$	0.461
2013-09-05	11:03:16	11.05	$24 + 11.05 - 17.89 = 17.17$	0.715
2014-04-10	01:15:33	1.26	$48 + 1.26 - 17.89 = 31.37$	1.307
2014-07-05	07:00:53	7.01	$48 + 7.01 - 17.89 = 37.12$	1.547
2014-10-27	14:08:50	14.15	$48 + 14.15 - 17.89 = 44.26$	1.844

There are six unknowns defining the sinusoidal curve: x_0, y_0, v_x, v_y, a, b . Fortunately, having taken pictures at six different times, we have six (x, y) pairs of data points that are supposed to lie on this curve. In other words, we have twelve equations and six unknowns. That’s a few more equations than is needed. We could drop the “extra” equations and then just try to solve for the six unknowns using the six remaining equations. A better approach is it try to “fit” all

twelve equations by minimizing the sum of the absolute values of the errors between the model formula and the measured centroids. In fact, there are very nice online tools that allow one to solve problems of this type. One of these online tools is located at a website called *NEOS: The Network Enabled Optimization Server* (<http://www.neos-server.org/neos/>). The NEOS website expects the user to upload a problem expressed in a language called AMPL (Fourer et al. (1993)). The AMPL language is a high-level language that is easy to read and fairly simple to learn. The entire AMPL formulation of the problem we wish to solve is shown in Figure 4. In the AMPL model, I used negative values for vertical y coordinates. This is because these pixels values increase as one moves down the picture which is backwards from how one normally works in mathematics.

The last few lines of the AMPL model print out the parallax and the distance to Barnard’s star in parsecs. The resolution in my imaging setup is 0.575 arcseconds/pixel. Hence, the parallax parameter a is multiplied by 0.575 to convert from pixels to arcseconds. Here’s the output from the AMPL model:

```
x0 = 680.745
y0 = -648.067
vx = -0.0687044
vy = -0.761649
a = 0.949463
b = 0.510165

orbit inclination = 28.250000 degrees
parallax = 0.545941
Distance to Barnards star = 1.831699 parsec
Proper motion = 10.553429 arcsec/yr
```

The distance I found, 1.832 parsecs, is just a tad smaller than the accurately known value of 1.835 ± 0.001 parsecs.

The computed parameters were used to make the overlaid sinusoidal plot shown in Figure 3. Clearly, the computed sinusoid matches the six data points very well. Figure 5 shows how Barnard’s star would wobble if it had zero proper motion.

The simple AMPL model was further extended to do a *bootstrap* analysis to produce a one-sigma confidence interval for the distance:

$$\text{Distance} = 1.832 \pm 0.061 \text{ parsecs.}$$

Finally, note that the above best-fit calculation also gives a quantitative value for the magnitude of the velocity of Barnard’s star in the plane perpendicular to our line of sight. Specifically,

```
param pi := 4*atan(1);
param n := 6;
param x {1..n};
param y {1..n};
param t {1..n};
var x0;
var y0;
var vx;
var vy;
var a;
var b = a*tan(28.25*pi/180); # Barnard's star is 28.25 degrees above ecliptic
var eps {1..n} >= 0;
var delta {1..n} >= 0;
let x[1] := 681.527; let y[1] := -639.539;
let x[2] := 680.374; let y[2] := -656.999;
let x[3] := 678.639; let y[3] := -661.234;
let x[4] := 679.479; let y[4] := -672.151;
let x[5] := 677.950; let y[5] := -676.775;
let x[6] := 676.727; let y[6] := -681.713;
# time measured with 1 year = 24 "hours"
let t[1] := 6.024 - 17.890;
let t[2] := 24 + 4.967 - 17.890;
let t[3] := 24 + 11.054 - 17.890;
let t[4] := 48 + 1.259 - 17.890;
let t[5] := 48 + 7.015 - 17.890;
let t[6] := 48 + 14.147 - 17.890;

minimize sumdev: sum {j in 1..n} (eps[j] + delta[j]) ;
subject to eps_def_lo {j in 1..n}: -eps[j] <= x0 + vx*t[j] + a*sin(2*pi*t[j]/24) - x[j];
subject to eps_def_hi {j in 1..n}: x0 + vx*t[j] + a*sin(2*pi*t[j]/24) - x[j] <= eps[j];
subject to del_def_lo {j in 1..n}: -delta[j] <= y0 + vy*t[j] + b*cos(2*pi*t[j]/24) - y[j];
subject to del_def_hi {j in 1..n}: y0 + vy*t[j] + b*cos(2*pi*t[j]/24) - y[j] <= delta[j];
solve;
display x0, y0, vx, vy, a, b;
printf "orbit inclination = %f degrees \n", atan(b/a)*180/pi;
printf "parallax = %f \n", a*0.575;
printf "Distance to Barnards star = %f parsec \n", 1/(a*0.575);
printf "Proper motion = %f arcsec/yr \n", sqrt(vx^2 + vy^2)*0.575*24;
```

Fig. 4.— The AMPL model to find the best-fitting sinusoid.




Fig. 5.— This plot shows the data points, the solution to the full regression model and the solutions one would obtain if Barnard’s star had either no parallax or no proper motion.

the measured proper motion is 10.553 arcseconds per year. This value matches quite well with the established rate of proper motion of 10.37 arcseconds per year. And, the velocity in arcseconds per year times the distance in parsecs gives the velocity in astronomical units (au) per year. Using our derived numbers we get that the projected velocity is 19.3 au/yr.

A natural question to ask is: why use “least absolute deviations” (LAD) instead of the more common procedure of minimizing the sum of the squares? We tried it both ways. Minimizing the sum of the squares, the computed parallax comes to 1.789 parsecs. This answer isn’t bad but it’s further from the known answer than the LAD calculation. In general, minimizing the sum of the absolute deviations provides an estimate that is more robust against outliers in the data. The fact that using a sum of squares gives a worse answer suggests that one or two of the measured values is somehow less accurate than the others. That is, there are a few outliers in the data.

Another question one might ask is: why use just two background stars rather than all of the background stars? Wouldn’t the latter give a better computation for aligning the images? Max-ImDL provides an option to align images using “auto star matching”. This method was tried first. In fact, for the first five images, this method gives a very good result. But, adding the sixth night’s image to the data, the result degraded dramatically. At first this was very puzzling. But, one needs to worry about the fact that at low altitude stars of different colors are refracted differently by the atmosphere. I had taken the images using a broad-band luminance filter and so this refraction issue could be a problem especially given that the sixth image was acquired when Barnard’s star was somewhat closer to the horizon than on all previous nights. Also, one needs to be careful about images closer to the horizon being compressed somewhat in north-south direction relative to east-west and that this could affect the star centroiding algorithm. These two effects are well-known and usually can be ignored. But, given that this experiment relies on computing positions with a precision on the order of 1/100-th of a pixel, such issues can be very important. And, it seems that they are.

Finally, note that in a previous paper (Vanderbei and Belikov (2007)) Rus Belikov and I made parallax measurements on the asteroid Prudentia and in so doing were able to measure the astronomical unit in terms of Earth radii. We discovered that

$$1 \text{ au} = 23,222 \text{ Earth radii.}$$

And, in another previous paper (Vanderbei (2008)) I measured the radius of the Earth by analyzing the compressed reflection of the Sun on the smooth surface of Lake Michigan during a sunset on a calm summer evening. The result so-obtained was

$$1 \text{ Earth radius} = 5200 \text{ miles} = 8400 \text{ km.}$$

These numbers aren’t terribly accurate but they are in the right ballpark. Anyway, combining these

measurements, we see that the distance to Barnard’s star is

$$\begin{aligned} 1.832 \text{ parsec} &= 1.832/\text{arcsec} \times \frac{3600 \text{ arcsecs}}{\text{degree}} \times \frac{180 \text{ degree}}{\pi \text{ radians}} \times 1 \text{ au} \\ &= 377,877 \text{ au} \times 23,222 \frac{\text{Earth radii}}{\text{au}} \\ &= 8,775,000,000 \text{ Earth radii} \times \frac{8400 \text{ km}}{\text{Earth radius}} \\ &= 73.7 \times 10^{12} \text{ km}. \end{aligned}$$

This final answer is pretty close to the known distance of 56.6 trillion kilometers.

Lastly, the proper motion translates to a projected velocity of

$$\begin{aligned} 19.3 \frac{\text{au}}{\text{yr}} &= 19.3 \frac{\text{au}}{\text{yr}} \times \frac{23,222 \text{ Earth radii}}{\text{au}} \times \frac{8400 \text{ km}}{\text{Earth radius}} \times \frac{1 \text{ yr}}{365.25 \text{ days}} \times \frac{1 \text{ day}}{24 \text{ hr}} \times \frac{1 \text{ hr}}{3600 \text{ sec}} \\ &= 120 \frac{\text{km}}{\text{sec}}. \end{aligned}$$

REFERENCES

- R. Fourer, D.M. Gay, and B.W. Kernighan. *AMPL: A Modeling Language for Mathematical Programming*. Duxbury Press, Belmont CA, 1993.
- R.J. Vanderbei. The Earth Is Not Flat: An Analysis of a Sunset Photo. *Optics and Photonics News*, November:34–39, 2008.
- R.J. Vanderbei and R. Belikov. Measuring the astronomical unit from your backyard. *Sky and Telescope*, 113(1):91–94, 2007.